


## Ricochet

00000 RICOCHET 1 00000



After solving the display the coding was done in no time. Again this game is based on one of my ONELINERS. The display is only moved. Controls are now up / down. Also some speedingup is added to the game. For fun I added my (slightly altered) logo as block.

```
; Ricochet, game 45 in 1k hires on the ZX81
```

```
? * TORNADO *
```

```
ORG #4009  
DUMP 49161
```

```
basic LD B,5 ; preset for 48K bug  
JR init0
```

```
DEFB 236,212,28 ; The BASIC  
DEFB 126 ; fully placed over sysvar  
DEFB 143,0,18 ; start to BASIC=#4009
```

```
eline DEFW last ; needed by loading  
chadd DEFW last-1  
xptr DEFW 0  
stkbot DEFW last  
stkend DEFW last  
berg DEFB 0  
mem DEFW 0  
DEFB 128
```

```
init1 JP init ; init can be anywhere
```

```

; all above reusable AFTER loading

lastk      DEFB 255,255,255      ; used by ZX81
margin     DEFB 55               ; used by ZX81
nxtlin     DEFW basic           ; reusable after load

init0      XOR  A                ; delay intrupts by
          DEFB 254              ; CP n ; skip flagx
flagx      DEFB 0

          EX  AF,AF'            ; intruptcounter reset
          DEFB #3A              ; LD A,(nn) ; skip taddr

taddr      DEFW 3213             ; used by ZX81
          LD  E,L               ; low byte equal 48K bug
          DEFB #3A              ; LD A,(NN) ; skip frames

frames     DEFW 65535            ; used by ZX81
coords     JR  init1            ; useable
prcc       DEFB 188             ; used by ZX81
sposn      DEFB 33,24           ; used by ZX81
cdflag     DEFB 64              ; used by ZX81

; when not shot, first byte = 0, exit HR
bullitscr  DEFB 0               ; ypos line of bullit

; BASIC initialization done on the screen
init       LD  IX,hr            ; Hires mode
          LD  SP,#4400
          LD  H,#3F             ; #3fxx
          LD  D,#BF             ; #bfxx
          LDIR                    ; repair 48K bug
          LD  HL,#4010

line2      DEFB 0               ; line 2 of bullit

er         DEC  L               ; reused to write offscreen
          LD  (HL),C
          JR  NZ,er             ; erase dataline

; redefine keys
          LD  HL,score          ; key to redefine
          LD  DE,init0          ; table of defined keys
redef      SET  7,(HL)          ; signal which key
          LD  A,(lastk)
          INC  A
          JR  NZ,redef          ; wait for release
waitkey    LD  BC,(lastk)
          LD  A,C
          INC  A
          JR  Z,waitkey        ; wait for keypress
          PUSH HL
          PUSH DE
          CALL #7BD             ; translate to ascii
          POP  DE
          POP  HL
          LD  (DE),A            ; save key
          INC  DE

          XOR  A
          LD  (HL),A            ; erase old direction
          INC  HL
          OR   (HL)
          JR  NZ,redef          ; test end reached

```

```

JP      start                ; start game

xposudg  DEFB 10,18          ; target
          DEFB 11,18
          DEFB 12,18
          DEFB 13,18
          DEFB 14,18
          DEFB 15,18

          DEFB 0,5           ; block
          DEFB 0,5
          DEFB 0,5
          DEFB 0,5
          DEFB 0,5
          DEFB 0,5

          DEFB 15,18         ; pistol
          DEFB 14,18
          DEFB 13,18
          DEFB 12,18
          DEFB 11,0
          DEFB 10,18

udgstack  DEFW pistol
          DEFW pistol
          DEFW pistol
          DEFW pistol
          DEFW pistol
          DEFW pistol

          DEFW block
          DEFW block
          DEFW block
          DEFW block
          DEFW block
          DEFW block

          DEFW target
          DEFW target
          DEFW target
          DEFW target
          DEFW target
          DEFW target

          DEFW savesp        ; extra fake POP HL
          DEFW savesp        ; RET on bottom screen

pistol    DEFB 95,128,63,64,104,32,112,16
          DEFB 96,8,96,4,0,2,0,0 ; pistol and "\"

block     DEFB 20,87,20,84,20,114,20,25
          DEFB 20,119,20,24,0,20,0,0 ; YRS logo and line

target    DEFB 30,2,33,4,45,8,45,16
          DEFB 33,32,30,64,0,128,0,0 ; target and "/"

hr        LD  HL,lowres+#8000  ; the lowres display
          LD  BC,#311
          LD  A,#1E
          LD  I,A
          LD  A,#FB
          CALL #2B5

```

```

hr00      LD    B,17                ; sync hires display
          DJNZ  hr00

          LD    (savesp+1),SP      ; save SP
          SET  0,(HL)              ; timing 15 tstates in 2 bytes
          LD    SP,udgstack        ; use stack for udgs

          LD    HL,lbuf+#8000      ; the display line
          LD    DE,bullitscr
          LD    BC,#9108           ; the screensize

          EXX
          LD    BC,udgstack-1
          LD    D,#40
          LD    A,D
          LD    I,A                ; all display data in #40..

          EXX

bloop     DEC  B
          LD    C,8
          EXX
          POP  HL                  ; get next udg pointer

          EXX
          LD    A,(DE)
          SUB  B
          JP   NZ,nline2           ; test display bullit

          OR   B
          RET  Z                   ; test end of screen

bullit    EXX
          DEC  C
          INC  L
          LD   A,(BC)
          LD   E,A
          LDI                      ; clear udg here
          EXX
          LD   A,bullitscr*256/256+1
          LD   E,line2*256/256     ; signal to end of screen
          DEC  C
          JP   (HL)                ; display line

bullit2   EXX                    ; needed, 4 tstates too soon
          DEC  BC                  ; correct the 4 tstates
          INC  HL
          LD   A,(BC)
          LD   E,A
          LDI                      ; again clear udg
          EXX
          LD   E,line2*256/256
          LD   A,bullitscr*256/256+1
          DEC  C
          JP   (HL)                ; display line

cloop     DEC  B
          EXX
          INC  C                  ; undo change udgpointers
          INC  BC                  ; for timing BC and C

nline     EXX
          LD   A,(DE)              ; get bullit y
          SUB  B

```

```

JR      Z,bullit2          ; do bullitscreen

nline2  EXX
        LD      A,(BC)      ; get x udg
        LD      E,A        ; set to DEstination
        LDI     ; copy udg
        LD      A,(BC)
        LD      E,A
        LDI     ; copy udg

        XOR     A          ; point to start of data
        EXX
        DEC     C
        JP      (HL)       ; display line

lbuf    LD      R,A
        DEFW    #8080,#8080,#8080 ; inverted display
        DEFW    #8080,#8080,#8080
        DEFW    #8080,#8080
        JP      Z,bloop     ; 48K bug
        JP      cloop      ; 48K bug

savesp  LD      SP,0        ; restore SP in this game
        CALL    #292        ; back from intrupt
        CALL    #220
        LD      IX,hr
        JP      #2A4

eog     LD      HL,score-1   ; test for hiscore
        LD      DE,hiscore-1
        LD      BC,6

same    INC     HL
        INC     DE
        DEC     C
        JR      Z,start
        LD      A,(DE)
        CP      (HL)
        JR      Z,same
        CALL    C,#19F9

start   LD      A,%10111111   ; game over, wait for
        IN      A,(254)
        RRA          ; newline
        JR      C,start

        LD      HL,score     ; erase old score
        LD      B,5
        LD      (HL),28
        INC     HL
        DJNZ    clsc

        LD      A,160        ; reset speed up
        LD      (speed+1),A

        LD      HL,lives     ; set 3 lives
        LD      (HL),32
        DEFB    33          ; skip dead

dead    LD      (HL),18      ; erase target
        LD      HL,lives
        DEC     (HL)
        LD      A,28
        CP      (HL)
        JR      Z,eog       ; end of game

```

```

rseed      LD    DE,0
           LD    HL,(frames)
           ADD   HL,DE
           DEC   HL
           LD    A,H
           AND   #1F
           LD    H,A
           LD    (rseed+1),HL
           LD    A,(HL)
frnd        LD    B,A
           SUB   6
           JR    NC,frnd
           INC   B

fnew        LD    HL,xposudg-1
           INC   HL
           INC   HL
           DJNZ  fnew

           LD    (HL),B           ; set target on start

           LD    A,200
           LD    (frames),A       ; reset timer

ploop       XOR   A
           LD    (bullitscr),A    ; do not show bullit

           CALL  readkey          ; d = udlrf...

ypistol     LD    B,4
           BIT   7,D
           JR    NZ,moveup
           BIT   6,D
           JR    Z,testb
movedown    INC   B
           INC   B
moveup      DEC   B
           JR    Z,movedown
testb       LD    A,B
           CP    7
           JR    Z,moveup

           LD    (ypistol+1),A    ; save new y position

           LD    C,6
           LD    HL,udgstack+1
erpist      DEC   HL
           DEC   HL
           LD    (HL),18          ; out of screen
           DJNZ  clpist          ; clear old pistols
           LD    (HL),B          ; visible position pistol
clpist      DEC   C
           JR    NZ,erpist

           BIT   3,D
           JP    Z,deadtest      ; no fire pressed

; now do a shot
           ADD   A,A              ; translate ypos to a line
           ADD   A,A
           ADD   A,A
           LD    C,A
           LD    A,#97

```

```

SUB    C

LD     B,A           ; ypos in B
LD     C,11

LD     DE,1          ; dy=0 , dx=1

; bullitloop here

bulloop LD    A,C           ; do dx
        ADD   A,E
        LD    C,A

        LD    A,B           ; do dy
        ADD   A,D
        LD    B,A

        LD    HL,bullitscr   ; clear bullitscreen
        LD    A,20
        LD    (HL),B         ; y pos set
clrbul  INC    HL
        LD    (HL),0         ; clear bullit line
        DEC   A              ; and also signal end screen
        JR    NZ,clrbul

; set correct bullitscreen
; find correct UDG and set these on matching x

        LD    A,B
        DEC   A              ; -1

        RRCA                ; /8
        RRCA
        RRCA
        AND   31             ; int
        INC   A              ; +1

        PUSH  DE
        LD    HL,pistol-4
        LD    DE,xposudg-2

fudg    DEC   HL
        DEC   HL
        INC   DE
        INC   DE
        DEC   A
udg2    JR    NZ,fudg

        PUSH  BC
        LD    A,B
        DEC   A

        LD    C,(HL)
        INC   HL
        LD    B,(HL)         ; BC=udgdata
        LD    HL,17
        ADD   HL,BC
        LD    B,H
        LD    C,L           ; now at end of udgdata

        AND   7
        INC   A

fpoint  DEC   BC

```

```

DEC BC
DEC A
JR NZ,fpoint          ; correct data within udg

udgpoint LD HL,bullitscr+1

        PUSH HL

        LD A,(DE)      ; get xpos
        ADD A,L
        LD L,A
        LD A,(BC)
        LD (HL),A      ; draw udg on bullit line

        POP HL
        PUSH HL
        INC DE
        DEC BC

        LD A,(DE)
        ADD A,L
        LD L,A
        LD A,(BC)
        LD (HL),A      ; draw second udg

        POP HL

; now plot
        POP BC
        POP DE

        LD A,C
        RRCA
        RRCA
        RRCA
        AND 15

        INC A

findx   INC HL
        DEC A
        JR NZ,findx
        DEC HL

        LD A,C
        AND 7
        INC A

        PUSH BC

        LD B,A
        LD A,1
fbit    RRCA
        DJNZ fbit

        LD C,(HL)      ; get screen info
        LD (HL),A      ; plot 1 bit only

        SUB C

        JR NZ,testc    ; no collission yet

        SUB E          ; ricochet of bullit
        LD E,D         ; dx=dy

```



```

LD    D,A                ; dy=-dx

testc    LD    A,C
         CP    30          ; test target hit?

wfr2     LD    B,50        ; some delay to see
         DJNZ  wfr2        ; the bullit go

         POP    BC         ; get original x y

         JR    Z,hit       ; target is hit

         LD    A,C
         CP    3
         JP    NC,bulloop   ; continue untill missed

deadtest CALL  ftarget
         LD    A,(frames)
speed    CP    160
         JR    NC,wait
         INC   (HL)        ; move up timer step
         LD    A,200
         LD    (frames),A   ; reset timer
         LD    A,(HL)
         CP    5
         JP    NC,dead     ; test missed

wait     LD    HL,frames   ; stable gameplay
         LD    A,(HL)
         SUB   4
wfr      CP    (HL)
         JR    NZ,wfr
         JP    ploop       ; continue playloop

hit      CALL  ftarget     ; handle score
         LD    A,5
         SUB   (HL)        ; 5-xpos = points
         LD    (HL),18     ; clear hit target
         LD    B,A
         LD    DE,(score+2) ; get 100-pos
nextb    LD    HL,score+5
         DEFB  #3A         ; hide ten
ten      LD    (HL),28
         DEC   HL
         INC   (HL)
         LD    A,(HL)
         CP    38
         JR    Z,ten
         DJNZ  nextb
         LD    A,(score+2)
         CP    E           ; test 100 points
         JR    Z,notup     ; if changed, speed up
         LD    A,(speed+1)
         ADD   A,5
         CP    190
         JR    Z,notup     ; not over max speed
         LD    (speed+1),A
notup    JP    rseed       ; get new random start

ftarget  LD    HL,xposudg+1 ; pointer target x
         LD    B,6
foundt   LD    A,(HL)
         SUB   5
         RET   C

```

```

        INC    HL
        INC    HL
        DJNZ   foundt
        RET

readkey    LD     BC,(lastk)
           LD     A,C
           INC    A
           JR     Z,zxpanic      ; no key pressed

           CALL   #7BD          ; translate to ascii

           LD     HL,init0      ; keytable
           LD     D,0
           CP     (HL)          ; test up
           INC    HL
           JR     NZ,testdown
           SET    7,D            ; signal up

testdown   CP     (HL)          ; test down
           INC    HL
           JR     NZ,testfire
           SET    6,D            ; signal down

testfire   CP     (HL)          ; test fire
           RET     NZ
           SET    3,D            ; signal fire
           RET                ; false key will skip zxpanic

zxpanic    LD     BC,%1110000000000111
           LD     A,#A0
           OUT    (C),A
           JR     zxw

zxw        IN     A,(C)
           CPL
           LD     D,A            ; udlfr... from zxpanic
           RET

x          EQU    101
n          EQU    27

lowres     DEFB   118,0,0,0,0,0,0
score      DEFB   "U"-n,"D"-n,"F"-n ; UDF for define keys
           DEFB   0,0,0
           DEFB   "R"+x,"I"+x,"C"+x,"O"+x,"C"+x
           DEFB   "H"+x,"E"+x,"T"+x,0

lives      DEFB   28,0
hiscore    DEFB   28,28,28,28,28
           DEFB   118,118

vars       DEFB   128
?
last       EQU    $

```